Pensieve header: A program to fix UO sequences in virtual tangle diagrams (old).

```
In[•]:= SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\OU"]
```

```
Out[•]= C:\drorbn\AcademicPensieve\Projects\OU
```

pdf

```
In[•]:= SetAttributes[VD, Orderless]
```

```
In[•]:= vd = VD[X_-1[4, 1], EOS[5], X_-1[3, 6], X_-1[7, 2], EOS[8]]
```

```
Out[•]= VD[EOS[5], EOS[8], X_-1[3, 6], X_-1[4, 1], X_-1[7, 2]]
```

```
In[•]:= js = Cases[vd, X_[_, j_] :> j] ⋂ Cases[vd, X_[i_, _] :> i - 1]
```

```
Out[•]= {2, 6}
```

```
In[•]:= j1 = First[js]
```

```
Out[•]= 2
```

```
In[•]:= {{s1, i1, j1}} = Cases[vd, X_s_[i_, j1] :> {s, i, j1}]
```

```
Out[•]= {{-1, 7, 2}}
```

```
In[•]:= {{s2, i2, j2}} = Cases[vd, X_s_[j1 + 1, j_] :> {s, j1 + 1, j}]
```

```
Out[•]= {{-1, 3, 6}}
```

```
In[•]:= Complement[vd, VD[X_s1[i1, j1], X_s2[i2, j2]]]
```

```
Out[•]= VD[EOS[5], EOS[8], X_-1[4, 1]]
```



```
In[•]:= out = Union[
  Complement[vd, VD[X_s1[i1, j1], X_s2[i2, j2]]],
  VD[X_s2[j1, j2], X_s1[i1, i2], X_s1 s2[i1 - .3 s1, j2 + .3 s2], X_-s1 s2[i1 + .3 s1, j2 - .3 s2]]
 ]
```

```
Out[•]= VD[EOS[5], EOS[8], X_-1[2, 6], X_-1[4, 1], X_-1[6.7, 6.3], X_-1[7, 3], X_1[7.3, 5.7]]
```

pdf

```
In[•]:= Tidy[vd_VD] := Module[{ps = Union @@ (List @@@ vd)},
  Replace[vd, Thread[ps → Range@Length@ps], {2}] ]
```

```
In[•]:= Tidy[out]
Out[•]= VD[EOS[5], EOS[12], X_-1[2, 7], X_-1[4, 1], X_-1[9, 8], X_-1[10, 3], X_1[11, 6]]
```

pdf

```
In[•]:= R12Reduce1[vd_VD] := Tidy@Module[{R2s, R2}, Which[
    Length[R2s = Cases[vd, X_s_[i_, j_] :> X_-s[i + 1, j + 1]] ⋂ (List @@ vd)] > 0,
    Complement[vd, VD[R2 = First@R2s, R2 /. X_s_[i_, j_] :> X_-s[i - 1, j - 1]]],
    Length[R2s = Cases[vd, X_s_[i_, j_] :> X_-s[i + 1, j - 1]] ⋂ (List @@ vd)] > 0,
    Complement[vd, VD[R2 = First@R2s, R2 /. X_s_[i_, j_] :> X_-s[i - 1, j + 1]]],
    True, DeleteCases[vd, X_[i_, j_] /; Abs[i - j] == 1] ]]
```

pdf

```
In[•]:= R12Reduce[vd_VD] := FixedPoint[R12Reduce1, vd]
```

pdf

```
In[•]:= γ[vd_VD] := Module[{js, s1, i1, j1, s2, i2, j2},
  js = Cases[vd, X_[_, j_] :> j] ⋂ Cases[vd, X_[i_, _] :> i - 1];
  If[Length[js] == 0, vd,
   j1 = RandomChoice[js]; i2 = j1 + 1;
   Cases[vd, X_s_[i_, j1] :> (s1 = s; i1 = i)];
   Cases[vd, X_s_[i2, j_] :> (s2 = s; j2 = j)];
   Tidy@Join[Complement[vd, VD[X_s1[i1, j1], X_s2[i2, j2]]],
     VD[X_s2[j1, j2], X_s1[i1, i2], X_s1 s2[i1 - s1/3, j2 + s2/3], X_-s1 s2[i1 + s1/3, j2 - s2/3]]
   ] ]]
```

pdf

```
In[•]:= Γ[vd_VD] := FixedPoint[γ, vd, 2^8]
```

pdf

```
In[•]:= Γ[T_] /; Head[T] =!= VD := Γ[VD[T]]
```

pdf

```
In[•]:= Γ̄[vd_VD] := FixedPoint[γ @* R12Reduce, vd, 2^8]
```

pdf

```
In[•]:= Γ̄[T_] /; Head[T] =!= VD := Γ̄[VD[T]]
```

```
In[•]:= vd
Out[•]= VD[EOS[5], EOS[8], X_-1[3, 6], X_-1[4, 1], X_-1[7, 2]]

In[•]:= vd // γ
Out[•]= VD[EOS[5], EOS[12], X_-1[2, 7], X_-1[4, 1], X_-1[9, 8], X_-1[10, 3], X_1[11, 6]]

In[•]:= vd // γ // γ
Out[•]= VD[EOS[13], EOS[16], X_-1[5, 5], X_-1[7, 6], X_-1[8, 8],
  X_-1[10, 15], X_-1[12, 1], X_-1[14, 3], X_1[9, 4], X_1[11, 2]]
```

```
In[•]:= vd // γ // γ // γ
```

```
Out[•]= VD[EOS[17], EOS[20], X_1[5, 5], X_1[7, 7], X_1[9, 8], X_1[10, 10],
         X_1[12, 12], X_1[14, 19], X_1[16, 1], X_1[18, 3], X_1[11, 6], X_1[13, 4], X_1[15, 2]]
```

```
In[•]:= vd // γ // γ // γ // γ
```

```
Out[•]= VD[EOS[19], EOS[24], X_1[3, 7], X_1[11, 10], X_1[12, 12], X_1[13, 5], X_1[16, 23],
         X_1[18, 1], X_1[20, 8], X_1[21, 4], X_1[14, 7], X_1[15, 9], X_1[17, 2], X_1[22, 6]]
```

```
In[•]:= vd // Γ // Short
```

```
Out[•]//Short= VD[EOS[1017], EOS[1032], X_1[1, 30], X_1[18, 73],
                <<524>>, X_1[1020, 132], X_1[1022, 60], X_1[1025, 23], X_1[1026, 78]]
```

pdf

```
In[•]:= VPB[n_, {σs___}] := VPB[n, σs];
```

pdf

```
In[•]:= VD /: vd1_VD ** vd2_VD := Module[{es1, es2, m2},
           es1 = Cases[vd1, EOS[i_] :> i];
           m2 = Max[es2 = Cases[vd2, EOS[i_] :> i]];
           Tidy[vd1 ⋃ Replace[DeleteCases[vd2, _EOS],
              i_ :> i / m2 - 1 + es1[[1 + Count[es2, e_ /; i > e]]], {2}]]
         ]
```

```
In[•]:= vd
```

```
Out[•]= VD[EOS[5], EOS[8], X_1[3, 6], X_1[4, 1], X_1[7, 2]]
```

```
In[•]:= vd ** vd
```

```
Out[•]= VD[EOS[9], EOS[14], X_1[3, 10], X_1[4, 1], X_1[7, 12], X_1[8, 5], X_1[11, 2], X_1[13, 6]]
```

```
In[•]:= vd ** vd ** vd
```

```
Out[•]= VD[EOS[13], EOS[20], X_1[3, 14], X_1[4, 1], X_1[7, 16],
         X_1[8, 5], X_1[11, 18], X_1[12, 9], X_1[15, 2], X_1[17, 6], X_1[19, 10]]
```

pdf

```
In[•]:= VD[VPB[n_]] := VD @@ (EOS /@ Range[n]);
        VD[VPB[n_, σ_{i_,j_}]] := Tidy@Append[VD @@ (EOS /@ Range[n]), X_{+1}[i - 0.5, j - 0.5]];
        VD[VPB[n_, σ̄_{i_,j_}]] := Tidy@Append[VD @@ (EOS /@ Range[n]), X_{-1}[i - 0.5, j - 0.5]];
        VD[VPB[n_, σ_, σs__]] := VD[VPB[n, σ]] ** VD[VPB[n, σs]]
```

```
In[•]:= VD[VPB[5, σ̄_{4,2}]]
```

```
Out[•]= VD[EOS[1], EOS[3], EOS[4], EOS[6], EOS[7], X_1[5, 2]]
```

```
In[•]:= vd1 = VD[VPB[5, σ_{2,3}]]
```

```
Out[•]= VD[EOS[1], EOS[3], EOS[5], EOS[6], EOS[7], X_1[2, 4]]
```

```
In[•]:= vd2 = VD[VPB[5, σ_{3,4}]]
```

```
Out[•]= VD[EOS[1], EOS[2], EOS[4], EOS[6], EOS[7], X_1[3, 5]]
```

```
In[•]:= VD[VPB[5, σ2,3, σ3,4]]
Out[•]= VD[EOS[1], EOS[3], EOS[6], EOS[8], EOS[9], X1[2, 4], X1[5, 7]]

In[•]:= VD[VPB[5, σ2,3, σ3,4]] // Γ
Out[•]= VD[EOS[1], EOS[5], EOS[8], EOS[12], EOS[13], X-1[4, 9], X1[2, 11], X1[3, 7], X1[6, 10]]

In[•]:= VPB[3, σ1,2, σ1,3, σ2,3] // Γ
Out[•]= VD[EOS[5], EOS[8], EOS[13], X-1[3, 10], X1[1, 12], X1[2, 7], X1[4, 9], X1[6, 11]]

In[•]:= VPB[3, σ2,3, σ1,3, σ1,2] // Γ
Out[•]= VD[EOS[3], EOS[6], EOS[9], X1[1, 8], X1[2, 5], X1[4, 7]]

In[•]:= R2ReduceB[vd_VD] := Module[{R2s, R2},
          R2s = Cases[vd, Xs_[i_, j_] :> X-s[i + 1, j + 1]] ⋂ (List @@ vd);
          If[Length[R2s] == 0, vd,
           R2 = First@R2s;
           Tidy@Complement[vd, VD[R2, R2 /. Xs_[i_, j_] :> X-s[i - 1, j - 1]]]
          ]]

In[•]:= R2ReduceC[vd_VD] := Module[{R2s, R2},
          R2s = Cases[vd, Xs_[i_, j_] :> X-s[i + 1, j - 1]] ⋂ (List @@ vd);
          If[Length[R2s] == 0, vd,
           R2 = First@R2s;
           Tidy@Complement[vd, VD[R2, R2 /. Xs_[i_, j_] :> X-s[i - 1, j + 1]]]
          ]]

In[•]:= R2Reduce[vd_VD] := FixedPoint[R2ReduceB @* R2ReduceC, vd]

In[•]:= R1Reduce1[vd_VD] := Tidy@DeleteCases[vd, X_[i_, j_] /; Abs[i - j] == 1]

In[•]:= VPB[3, σ1,2, σ1,3, σ2,3] // Γ // R2Reduce
Out[•]= VD[EOS[3], EOS[6], EOS[9], X1[1, 8], X1[2, 5], X1[4, 7]]

In[•]:= VPB[2, σ1,2, σ2,1] // VD
Out[•]= VD[EOS[3], EOS[6], X1[1, 4], X1[5, 2]]

In[•]:= VPB[2, σ1,2, σ2,1] // VD // Γ
Out[•]= VD[EOS[7], EOS[10], X-1[3, 4], X1[1, 6], X1[2, 9], X1[8, 5]]

In[•]:= VPB[2, σ1,2, σ2,1] // VD // Γ // R2Reduce
Out[•]= VD[EOS[7], EOS[10], X-1[3, 4], X1[1, 6], X1[2, 9], X1[8, 5]]

In[•]:= VPB[2, σ1,2, σ2,1] // VD // Γ // R12Reduce
Out[•]= VD[EOS[5], EOS[8], X1[1, 4], X1[2, 7], X1[6, 3]]
```

```
In[•]:= Test1[n_, m_] := Module[{gens, i, j, k, l},
          gens = Flatten@Table[{σ_{i,j}, σ̄_{i,j}}, {i, n}, {j, DeleteCases[Range@n, i]}];
          Table[
           {i, j, k} = ijk;
           Γ̄[VPB[n, Sequence @@ p, σ_{i,j}, σ_{i,k}, σ_{j,k}, Sequence @@ q]] ==
            Γ̄[VPB[n, Sequence @@ p, σ_{j,k}, σ_{i,k}, σ_{i,j}, Sequence @@ q]],
           {l, 0, m - 3}, {p, Tuples[gens, l]}, {q, Tuples[gens, m - 3 - l]},
           {ijk, Join @@ (Permutations /@ Subsets[Range[n], {3}])}
          ]]

In[•]:= Test1[3, 3]

Out[•]= {{{{True, True, True, True, True, True}}}}

In[•]:= Timing@Union@Flatten@Test1[4, 5]

Out[•]= {127.859, {True}}

In[•]:= Timing@Union@Flatten@Test1[5, 4]

Out[•]= {9.17188, {True}}
```

```
In[•]:= Test2[n_, m_] := Module[{gens, s, r = 0, ij, ijk, ijkl, perm, i, j, k, l, tests},
  gens = Flatten@Table[{σ_{i,j}, σ̄_{i,j}}, {i, n}, {j, DeleteCases[Range@n, i]}];
  tests = Flatten[{
     Table[{i, j} = ij; {
       T[VPB[n, Join[p, {σ_{i,j}, σ̄_{i,j}}, q]], VPB[n, Join[p, q]]],
       T[VPB[n, Join[p, {σ̄_{i,j}, σ_{i,j}}, q]], VPB[n, Join[p, q]]]
      },
      {s, 0, m - 2}, {t, 0, s}, {p, Tuples[gens, t]}, {q, Tuples[gens, s - t]},
      {ij, Join @@ (Permutations /@ Subsets[Range[n], {2}])}
     ],
     Table[{i, j, k} = ijk; {
       T[VPB[n, Join[p, {σ_{i,j}, σ_{i,k}, σ_{j,k}}, q]], VPB[n, Join[p, {σ_{j,k}, σ_{i,k}, σ_{i,j}}, q]]],
       T[VPB[n, Join[p, {σ̄_{j,i}, σ_{i,k}, σ_{j,k}}, q]], VPB[n, Join[p, {σ_{j,k}, σ_{i,k}, σ̄_{j,i}}, q]]],
       T[VPB[n, Join[p, {σ_{i,j}, σ_{i,k}, σ̄_{k,j}}, q]], VPB[n, Join[p, {σ̄_{k,j}, σ_{i,k}, σ_{i,j}}, q]]],
       T[VPB[n, Join[p, {σ_{i,j}, σ̄_{k,i}, σ̄_{k,j}}, q]], VPB[n, Join[p, {σ̄_{k,j}, σ̄_{k,i}, σ_{i,j}}, q]]],
       T[VPB[n, Join[p, {σ̄_{j,i}, σ̄_{k,i}, σ_{j,k}}, q]], VPB[n, Join[p, {σ_{j,k}, σ̄_{k,i}, σ̄_{j,i}}, q]]],
       T[VPB[n, Join[p, {σ̄_{j,i}, σ̄_{k,i}, σ̄_{k,j}}, q]], VPB[n, Join[p, {σ̄_{k,j}, σ̄_{k,i}, σ̄_{j,i}}, q]]]
      },
      {s, 0, m - 3}, {t, 0, s}, {p, Tuples[gens, t]}, {q, Tuples[gens, s - t]},
      {ijk, Join @@ (Permutations /@ Subsets[Range[n], {3}])}
     ],
     Table[{i, j, k, l} = ijkl[[perm]]; {
       T[VPB[n, Join[p, {σ_{i,j}, σ_{k,l}}, q]], VPB[n, Join[p, {σ_{k,l}, σ_{i,j}}, q]]],
       T[VPB[n, Join[p, {σ̄_{i,j}, σ_{k,l}}, q]], VPB[n, Join[p, {σ_{k,l}, σ̄_{i,j}}, q]]],
       T[VPB[n, Join[p, {σ_{i,j}, σ̄_{k,l}}, q]], VPB[n, Join[p, {σ̄_{k,l}, σ_{i,j}}, q]]],
       T[VPB[n, Join[p, {σ̄_{i,j}, σ̄_{k,l}}, q]], VPB[n, Join[p, {σ̄_{k,l}, σ̄_{i,j}}, q]]]
      },
      {s, 0, m - 2}, {t, 0, s}, {p, Tuples[gens, t]}, {q, Tuples[gens, s - t]},
      {ijkl, Subsets[Range[n], {4}]}, {perm, {{1, 2, 3, 4}, {1, 3, 2, 4}, {1, 4, 2, 3}}}
     ]
    }];
  Cases[tests, T[b1_, b2_] /; T̄[b1] =!= T̄[b2]]
 ]

In[•]:= Timing@Test2[3, 3]
Out[•]= {0.34375, {}}

In[•]:= Timing@Test2[3, 4]
Out[•]= {7.09375, {}}

In[•]:= Timing@Test2[3, 5]
Out[•]= {196.203, {}}

In[•]:= Timing@Test2[4, 2]
Out[•]= {0.015625, {}}
```

```
In[•]:= Timing@Test2[4, 3]
Out[•]= {1.45313, {}}

In[•]:= VPB[3, σ̄1,2, σ1,3, σ2,3] // Γ // R2Reduce
Out[•]= VD[EOS[5], EOS[8], EOS[13], X-1[2, 7], X-1[3, 12], X1[1, 10], X1[4, 9], X1[6, 11]]

In[•]:= VPB[3, σ2,3, σ1,3, σ̄1,2] // Γ // R2Reduce
Out[•]= VD[EOS[3], EOS[6], EOS[9], X-1[2, 5], X1[1, 8], X1[4, 7]]

In[•]:= AllVPBInvariants[n_, m_] := Module[{gens, k},
          gens = Flatten@Table[{σi,j, σ̄i,j}, {i, n}, {j, DeleteCases[Range@n, i]}];
          Flatten@
           Table[VPB[n, Sequence @@ p] → Γ̄@VPB[n, Sequence @@ p], {k, 0, m}, {p, Tuples[gens, k]}]]

In[•]:= AllVPBInvariants[2, 2] // Column
Out[•]= VPB[2] → VD[EOS[1], EOS[2]]
        VPB[2, σ1,2] → VD[EOS[2], EOS[4], X1[1, 3]]
        VPB[2, σ̄1,2] → VD[EOS[2], EOS[4], X-1[1, 3]]
        VPB[2, σ2,1] → VD[EOS[2], EOS[4], X1[3, 1]]
        VPB[2, σ̄2,1] → VD[EOS[2], EOS[4], X-1[3, 1]]
        VPB[2, σ1,2, σ1,2] → VD[EOS[3], EOS[6], X1[1, 4], X1[2, 5]]
        VPB[2, σ1,2, σ̄1,2] → VD[EOS[1], EOS[2]]
        VPB[2, σ1,2, σ2,1] → VD[EOS[5], EOS[8], X1[1, 4], X1[2, 7], X1[6, 3]]
        VPB[2, σ1,2, σ̄2,1] → VD[EOS[7], EOS[10], X-1[1, 4], X-1[8, 5], X1[2, 9], X1[3, 6]]
        VPB[2, σ̄1,2, σ1,2] → VD[EOS[1], EOS[2]]
        VPB[2, σ̄1,2, σ̄1,2] → VD[EOS[3], EOS[6], X-1[1, 4], X-1[2, 5]]
        VPB[2, σ̄1,2, σ2,1] → VD[EOS[7], EOS[10], X-1[2, 9], X-1[3, 6], X1[1, 4], X1[8, 5]]
        VPB[2, σ̄1,2, σ̄2,1] → VD[EOS[5], EOS[8], X-1[1, 4], X-1[2, 7], X-1[6, 3]]
        VPB[2, σ2,1, σ1,2] → VD[EOS[3], EOS[8], X1[1, 6], X1[4, 7], X1[5, 2]]
        VPB[2, σ2,1, σ̄1,2] → VD[EOS[3], EOS[10], X-1[1, 8], X-1[4, 7], X1[5, 2], X1[6, 9]]
        VPB[2, σ2,1, σ2,1] → VD[EOS[3], EOS[6], X1[4, 1], X1[5, 2]]
        VPB[2, σ2,1, σ̄2,1] → VD[EOS[1], EOS[2]]
        VPB[2, σ̄2,1, σ1,2] → VD[EOS[3], EOS[10], X-1[5, 2], X-1[6, 9], X1[1, 8], X1[4, 7]]
        VPB[2, σ̄2,1, σ̄1,2] → VD[EOS[3], EOS[8], X-1[1, 6], X-1[4, 7], X-1[5, 2]]
        VPB[2, σ̄2,1, σ2,1] → VD[EOS[1], EOS[2]]
        VPB[2, σ̄2,1, σ̄2,1] → VD[EOS[3], EOS[6], X-1[4, 1], X-1[5, 2]]
```

pdf

```
In[•]:= VPBGenerators[n_] :=
          VPBGenerators[n] = Flatten@Table[{σi,j, σ̄i,j}, {i, n}, {j, DeleteCases[Range@n, i]}];

In[•]:= VPBGenerators[5]
Out[•]= {σ1,2, σ̄1,2, σ1,3, σ̄1,3, σ1,4, σ̄1,4, σ1,5, σ̄1,5, σ2,1, σ̄2,1, σ2,3, σ̄2,3,
         σ2,4, σ̄2,4, σ2,5, σ̄2,5, σ3,1, σ̄3,1, σ3,2, σ̄3,2, σ3,4, σ̄3,4, σ3,5, σ̄3,5, σ4,1, σ̄4,1,
         σ4,2, σ̄4,2, σ4,3, σ̄4,3, σ4,5, σ̄4,5, σ5,1, σ̄5,1, σ5,2, σ̄5,2, σ5,3, σ̄5,3, σ5,4, σ̄5,4}
```

```
(*CountOUForms[n_,m_]:=Module[{k},
  Length@Union@Flatten@Table[
      R12Reduce@Γ@VPB[n,Sequence@@p], {k,0,m},{p,Tuples[VPBGenerators[n],k]}]]*)
```

pdf

```
ProudFollowers[n_, σ_i_,j_] := ProudFollowers[n, σ_i,j] = Module[{p, q, s},
    Flatten@{σ_i,j, σ_j,i, σ̄_j,i,
      Table[{σ_p,q, σ_q,p, σ̄_p,q, σ̄_q,p}, {p, {i, j}}, {q, Complement[Range[n], {i, j}]}],
      Table[{σ_p,q, σ̄_p,q},
       {p, Complement[Range[i + 1, n], {j}]}, {q, Complement[Range[n], {i, j, p}]}]
     }];
ProudFollowers[n_, σ̄_i_,j_] := ProudFollowers[n, σ̄_i,j] = ProudFollowers[n, σ_i,j] /. σ_i,j → σ̄_i,j
```

```
In[•]:= ProudFollowers[5, σ_2,3]
```

```
Out[•]= {σ_2,3, σ_3,2, σ̄_3,2, σ_2,1, σ_1,2, σ̄_2,1, σ̄_1,2, σ_2,4, σ_4,2, σ̄_2,4, σ̄_4,2, σ_2,5, σ_5,2, σ̄_2,5, σ̄_5,2, σ_3,1, σ_1,3, σ̄_3,1,
 σ̄_1,3, σ_3,4, σ_4,3, σ̄_3,4, σ̄_4,3, σ_3,5, σ_5,3, σ̄_3,5, σ̄_5,3, σ_4,1, σ̄_4,1, σ_4,5, σ̄_4,5, σ_5,1, σ̄_5,1, σ_5,4, σ̄_5,4}
```

```
In[•]:= ProudFollowers[5, σ̄_2,3]
```

```
Out[•]= {σ̄_2,3, σ_3,2, σ̄_3,2, σ_2,1, σ_1,2, σ̄_2,1, σ̄_1,2, σ_2,4, σ_4,2, σ̄_2,4, σ̄_4,2, σ_2,5, σ_5,2, σ̄_2,5, σ̄_5,2, σ_3,1, σ_1,3, σ̄_3,1,
 σ̄_1,3, σ_3,4, σ_4,3, σ̄_3,4, σ̄_4,3, σ_3,5, σ_5,3, σ̄_3,5, σ̄_5,3, σ_4,1, σ̄_4,1, σ_4,5, σ̄_4,5, σ_5,1, σ̄_5,1, σ_5,4, σ̄_5,4}
```

pdf

```
ProudVPBs[n_, 0] := {VPB[n]};
ProudVPBs[n_, 1] := VPB[n, #] & /@ VPBGenerators[n];
ProudVPBs[n_, m_] /; m > 1 := Flatten[
  ProudVPBs[n, m - 1] /. VPB[n, σs___, σ_] :> (VPB[n, σs, σ, #] & /@ ProudFollowers[n, σ])]
```

```
In[•]:= ProudVPBs[2, 2]
```

```
Out[•]= {VPB[2, σ_1,2, σ_1,2], VPB[2, σ_1,2, σ_2,1], VPB[2, σ_1,2, σ̄_2,1], VPB[2, σ̄_1,2, σ̄_1,2],
 VPB[2, σ̄_1,2, σ_2,1], VPB[2, σ̄_1,2, σ̄_2,1], VPB[2, σ_2,1, σ_2,1], VPB[2, σ_2,1, σ_1,2],
 VPB[2, σ_2,1, σ̄_1,2], VPB[2, σ̄_2,1, σ̄_2,1], VPB[2, σ̄_2,1, σ_1,2], VPB[2, σ̄_2,1, σ̄_1,2]}
```

```
In[•]:= ProudVPBs[3, 3]
```

```
Out[•]= {VPB[3, σ_1,2, σ_1,2, σ_1,2], VPB[3, σ_1,2, σ_1,2, σ_2,1], VPB[3, σ_1,2, σ_1,2, σ̄_2,1],
 VPB[3, σ_1,2, σ_1,2, σ_1,3], VPB[3, σ_1,2, σ_1,2, σ_3,1], VPB[3, σ_1,2, σ_1,2, σ̄_1,3],
 VPB[3, σ_1,2, σ_1,2, σ̄_3,1], ··· 1438 ···, VPB[3, σ̄_3,2, σ̄_1,2, σ_3,1],
 VPB[3, σ̄_3,2, σ̄_1,2, σ̄_1,3], VPB[3, σ̄_3,2, σ̄_1,2, σ̄_3,1], VPB[3, σ̄_3,2, σ̄_1,2, σ_2,3],
 VPB[3, σ̄_3,2, σ̄_1,2, σ_3,2], VPB[3, σ̄_3,2, σ̄_1,2, σ̄_2,3], VPB[3, σ̄_3,2, σ̄_1,2, σ̄_3,2]}

large output | show less | show more | show all | set size limit...
```

pdf

```
CountOUForms[n_, m_] := Module[{k},
  Length@Union@Flatten@Table[Γ̄@vpb, {k, 0, m}, {vpb, ProudVPBs[n, k]}]]
```

```
In[•]:= Timing@CountOUForms[2, 1]
Out[•]= {0., 5}

In[•]:= Timing@CountOUForms[2, 2]
Out[•]= {0., 17}

In[•]:= Timing@CountOUForms[2, 3]
Out[•]= {0.0625, 53}

In[•]:= Timing@CountOUForms[2, 4]
Out[•]= {0.28125, 161}

In[•]:= Timing@CountOUForms[2, 5]
Out[•]= {2.45313, 485}

In[•]:= Timing@CountOUForms[2, 6]
Out[•]= {25.1406, 1457}

In[•]:= FindSequenceFunction[{5, 17, 53, 161, 485, 1457}]
Out[•]= -1 + 2 × 3^#1 &

In[•]:= FindLinearRecurrence[{5, 17, 53, 161, 485, 1457}]
Out[•]= {4, -3}

In[•]:= Timing@CountOUForms[3, 1]
Out[•]= {0., 13}

In[•]:= Timing@CountOUForms[3, 2]
Out[•]= {0.046875, 145}

In[•]:= Timing@CountOUForms[3, 3]
Out[•]= {1.03125, 1561}

In[•]:= Timing@CountOUForms[3, 4]
Out[•]= {22.7813, 16 717}

In[•]:= Timing@CountOUForms[3, 5]
Out[•]= {533.859, 178 873}

In[•]:= Timing@CountOUForms[3, 6]
Out[•]= {14 058.3, 1 913 737}

In[•]:= 17038.5`^2 / 484.328125`
Out[•]= 599 409.
```

```
In[•]:= FindSequenceFunction[{13, 145, 1561, 16717, 178873, 1913737}]
Out[•]= FindSequenceFunction[{13, 145, 1561, 16717, 178873, 1913737}]

In[•]:= Timing@CountOUForms[4, 1]
Out[•]= {0., 25}

In[•]:= Timing@CountOUForms[4, 2]
Out[•]= {0.1875, 529}

In[•]:= Timing@CountOUForms[4, 3]
Out[•]= {6.8125, 10873}

In[•]:= Timing@CountOUForms[4, 4]
Out[•]= {261.844, 222289}

In[•]:= Timing@CountOUForms[4, 5]
Out[•]= {10540.5, 4540201}

In[•]:= 9002.375`^2 / 243.4375`
Out[•]= 332910.

In[•]:= {25, 529, 10873, 222289, 4540201}
Out[•]= {25, 529, 10873, 222289, 4540201}

In[•]:= Timing@CountOUForms[5, 1]
Out[•]= {0., 41}

In[•]:= Timing@CountOUForms[5, 2]
Out[•]= {0.484375, 1361}

In[•]:= Timing@CountOUForms[5, 3]
Out[•]= {27.25, 43121}

In[•]:= Timing@CountOUForms[5, 4]
Out[•]= {1572.84, 1351481}

In[•]:= 1459.640625`^2 / 24.515625`
Out[•]= 86905.8

In[•]:= {41, 1361, 43121, 1351481}
Out[•]= {41, 1361, 43121, 1351481}

In[•]:= Timing@CountOUForms[6, 1]
Out[•]= {0.015625, 61}
```

```
In[•]:= Timing@CountOUForms[6, 2]
Out[•]= {0.9375, 2881}

In[•]:= Timing@CountOUForms[6, 3]
Out[•]= {141.844, 127 021}

In[•]:= Timing@CountOUForms[6, 4]
Out[•]= {6921.03, 5 484 721}

In[•]:= FindSequenceFunction@{61, 2881, 127 021, 5 484 721}
Out[•]= FindSequenceFunction[{61, 2881, 127 021, 5 484 721}]

In[•]:= Timing@CountOUForms[7, 1]
Out[•]= {0.03125, 85}

In[•]:= Timing@CountOUForms[7, 2]
Out[•]= {2.67188, 5377}

In[•]:= Timing@CountOUForms[7, 3]
Out[•]= {250.484, 310 633}

In[•]:= {85, 5377, 310 633}
Out[•]= {85, 5377, 310 633}

In[•]:= Timing@CountOUForms[8, 1]
Out[•]= {0.03125, 113}

In[•]:= Timing@CountOUForms[8, 2]
Out[•]= {4.01563, 9185}

In[•]:= Timing@CountOUForms[8, 3]
Out[•]= {492.625, 668 081}

In[•]:= {113, 9185, 668 081}
Out[•]= {113, 9185, 668 081}

In[•]:= Timing@CountOUForms[9, 1]
Out[•]= {0.03125, 145}

In[•]:= Timing@CountOUForms[9, 2]
Out[•]= {5.73438, 14 689}

In[•]:= Timing@CountOUForms[9, 3]
Out[•]= {925.922, 1 307 233}
```

CountOUForms[n,1]:

```
In[•]:= n // FindSequenceFunction@{1, 5, 13, 25, 41, 61, 85, 113, 145} // Simplify // TeXForm
```

```
Out[•]//TeXForm= 2 n^2-2 n+1
```

CountOUForms[n,2]:

```
In[•]:= n // FindSequenceFunction@{1, 17, 145, 529, 1361, 2881, 5377, 9185, 14689} // Simplify //
         TeXForm
```

```
Out[•]//TeXForm= 2 n^4+4 n^3-18 n^2+12 n+1
```

CountOUForms[n,3]:

```
In[•]:= n // FindSequenceFunction@{1, 53, 1561, 10873, 43121, 127021, 310633, 668081, 1307233} //
          Simplify // TeXForm
```

```
Out[•]//TeXForm= \frac{1}{3} \left(4 n^6+36 n^5-2 n^4-546 n^3+1066 n^2-558 n+3\right)
```

```
In[•]:= n // FindSequenceFunction@{1, 53, 1561, 10873, 43121, 127021, 310633, 668081, 1307233}
```

$$\text{Out[•]=}\ \frac{1}{3}\left(3-558\,n+1066\,n^2-546\,n^3-2\,n^4+36\,n^5+4\,n^6\right)$$

```
In[•]:= 1/3 (3 - 558 n + 1066 n^2 - 546 n^3 - 2 n^4 + 36 n^5 + 4 n^6) /. n → 1
```

```
Out[•]= 1
```